2-1-92
E6841

NASA Technical Memorandum 105424

Microwave Beam Power Transmission at an Arbitrary Range

L.R. Pinero, J.L. Christian, Jr., and R.J. Acosta
Lewis Research Center
Cleveland, Ohio

February 1992

NASA

MICROWAVE BEAM POWER TRANSMISSION AT AN ARBITRARY RANGE

L.R. Pinero, J.L. Christian Jr. and R.J. Acosta
NASA Lewis Research Center
Cleveland, Ohio 44135

ABSTRACT

The power transfer efficiency between two circular apertures at an arbitrary range is obtained numerically. Two apertures can have generally different sizes and arbitrary taper illuminations. The effect of distance and taper illumination on the transmission efficiency are investigated for equal size apertures. The result shows that microwave beam power is more effective at close ranges, namely distances less than $2D^2$/lambda. Also shown was the power transfer efficiency increase with taper illumination for close range distances. A computer program was developed for calculating the power transfer efficiency at an arbitrary range.

INTRODUCTION

The problem of microwave beam power transmission between two aperture antennas have been studied extensively (for example refs. 1-5) in open literature. The purpose of this paper is to determine the effective range for reliable and efficient beam power transfer between two aperture antennas. The transmitting tangential aperture field is simulated with a tapered aperture field distribution. A numerical evaluation of Kirchoff and Helmotz vector diffraction (ref. 6) solution is used to calculate the electric and magnetic fields distribution at arbitrary observation distances. To obtain the transmission efficiency the power transmitted is compared to the power intercepted by receiving aperture antenna located at an arbitrary distance from the transmitting antenna (see Figure 1). A description and a copy of the program are included in appendixes A and B, respectively.

POWER TRANSFER EFFICIENCY

Let antennas A_1 and A_2 have generally dissimilar circular apertures of radii a_1 and a_2, respectively, as shown in Figure 1. The antennas are separated by a length d along a common centerline (z-axis). This axis is perpendicular to the aperture planes of A_1 and A_2. The antennas are assumed to be matched to their feeding lines. Let assume that the transmitting antenna radiates an amount of power P_t and the receiving antenna intercepts an amount of power P_r. The transmission efficiency is given as:

$$\eta = P_r / P_t \tag{1}$$

The transmitted power P_t can be expressed as:

$$P_t = .5 \iint \mathrm{RE}\ (\mathbf{E_t}(x,y,0) \times \mathrm{H}_t^*(x,y,0))\ \hat{e}_z\ dS \tag{2}$$

and the received power P_r can be expressed as:

$$P_r = .5 \iint \mathrm{RE}\ (\mathbf{E_r}(x,y,d) \times \mathbf{H}_r^*(x,y,d))\ \hat{e}_z\ dS \tag{3}$$

where the asterisk indicates the complex conjugate values, $\hat{e}_z$ is the unit vector in the z direction, $\mathbf{E_t}(x,y,d)$ is the tangential electric field component, and $\mathbf{H_t}(x,y,d)$ is the tangential magnetic component of the field at an arbitrary location. The components $\mathbf{H_t}(x,y,0)$, $\mathbf{E_r}(x,y,d)$, and $\mathbf{H_r}(x,y,d)$ can be expressed in terms of $\mathbf{E_t}(x,y,0)$ which uniquely determines the field in the entire half-space $z > 0$ (ref. 7). The electric and magnetic fields at z = 0 are assumed to have the following forms respectively:

$$\mathbf{E_t}(x,y,0) = E_0(C + (1 - C)(1 - (p / R)^2))^p\ x \tag{4}$$

$$\mathbf{H_t}(x,y,0) = (\hat{e}_z \times \mathbf{E_t}(x,y,0)) / Z_0\ y \tag{5}$$

where the parameter C, p, R are defined as:

$$C = 10^{-ET/20} \quad \text{, ET: edge taper in dB} \tag{6}$$

$$p = x^2 + y^2 \quad \text{, radius} \tag{7}$$

$$R \quad \text{, radius of aperture } A_1 \tag{8}$$

The E_r and H_r fields due to the above aperture distribution are given by:

$$E_r(x,y,d) = \iint E_t(x,y,0)\ e^{-jkr}\ ((1/r)(jK + 1/r)\ n.\ r)\ dS \quad (9)$$

$$H_r(x,y,d) = (e_z \times E_r(x,y,d)) / Z_0 \quad (10)$$

Figure 2 depicts all required parameters to evaluate equations (9) and (10). Since the observation points are in the near field of the aperture, equations (9) and (10) are general, do not simplify further. In order to compute the fields numerically, we divide the aperture into a finite number of grid points and the total field at the n^{th} observation point is the cartesian sum of the fields due to all grids on the transmitting aperture.

A computer program based on equation (9) and (10) was developed to compute the transmitting electric field and power flow at an arbitrary distance. The program listing and a sample run is included in Appendix I and II respectively.

DISCUSSION AND RESULTS

The effects of range and aperture tapering were studied for two identical aperture antennas. The apertures and relevant parameters are described as follows:

Transmitting aperture	A_1 = 0.7854 m^2 (Dia = 1 m)
Receiving aperture	A_2 = 0.7854 m^2 (Dia = 1 m)
Operating frequency	f = 3 GHz
Observation range	.1 FF < R < 1 FF (FF = $2D^2$/lam)
Far field range (FF)	FF = 20 m
Taper range	0Db,-10dB,-15dB,-20dB,-25dB,-30dB

For the purpose of this report we will consider 50% efficiency to be the minimum in which beam power transmission will be useful. Figure 3 shows the resulting power transfer efficiency as a function of distance when the two apertures are of the same area. This results indicates that beam power is most effective at closer ranges. Uniform aperture illumination was used for the transmitting antenna. Figure 4 and show the results obtained for a cases when the transmitting antenna is two times larger than the receiving aperture. Figure 5 shows the results obtained for cases when the receiving antenna is two times larger than the transmitting antenna. Basically this result shows by increasing the transmitting antenna size does not necessary increase the transmission efficiency. In the contrary increasing the receiving antenna size will increase the amount of power that can be collect. Figure 6 shows the effect of tapering the transmitting aperture on the beam power efficiency at a given distance. Figure 5 indicates a greater efficiency number can be obtained by properly tapering the transmitting aperture. Figure 6 and 7 shows the previous cases (figure 3 and 4) but with a -20 db taper level introduced in the transmitting aperture. In this case the useful range was increased by 20% .

In general when we consider beam power transmission we have to carefully consider two parameters, the transmitting aperture size and the taper required to efficiently transmit power from point to point. It is also assumed in our discussion that the received antennas will be mechanically steered, therefore scanning will not be an issue. The results presented indicates that beam power should be considered as an alternative for point to point power distribution.

APPENDIX A

DESCRIPTION OF PROGRAM

A computer program was designed to calculate the power efficiency between two circular apertures at an arbitrary range. The method of analysis is vector-wave diffraction approach. The main features in this program (BEAM FORTRAN) are the calculation of power transfer efficiency, transmit power density and receive power density. The inputs are modified according to the case inside the program BEAM FORTRAN. The following is a description of all the inputs parameters to the the program BEAM.

Input Parameters (BEAM FORTRAN)

LAMBDA	wavelenght in mts.
LF	spacing of computation grids
TAREA	transmit area in mts square
RAREA	receive area in mts square
TOTAREA	total area for calculating power radiated in mts square
DZ	fraction of far-field distance (2D**2/lambda)
ET	edge taper of transmit aperture fields

Example of Inputs and Results

INPUTS:

```
LAMBDA    =  .1
LF        =  .5
TAREA     =  0.7854
RAREA     =  0.7854
TOTAREA   =  1.5708
DZ        =  .1
ET        =   0.
```

RESULTS:

```
FRACTION OF 2*D**2/LAMBDA  =  0.100000024
DISTANCE BETWEEN APERTURES = 0.157079869E-02 KM
EDGE TAPER = 0.000000000E+00 DB
TRANSMITTING APERTURE AREA = 0.785399973     SQ. MTS
RECEIVING APERTURE AREA =  0.785399973   SQ. MTS
CAPTURE APERTURE AREA = 1.57079983  SQ. MTS
TOTAL POWER RADIATED = 0.243498638E-03 W
TOTAL POWER RECEIVED = 0.221619441E-03 W
EFFICIENCY PERCENT = 91.0146332  %
```

APPENDIX B

COMPUTER PROGRAM

```
      PROGRAM BEAM
******* PROGRAM DESCRIPTION *****
* THIS PROGRAM CALCULATES THE BEAM POWER EFFICIENCY OF TWO APERTURES
* AT AN ARBITRARY RANGE. THE INPUTS ARE THE WAVELENGTH (LAMBDA) IN
* METERS, RECEIVING AND TRANSMITTING GRID SPACING AND AREA IN SQ MTS.,
* EDGE TAPER IN DBS AND THE SEPARATION BETWEEN THE APERTURES AS A
* FRACTION OF 2D**2/LAMBDA.
******* VARIABLES *****
*****      LF - GRID SPACING
***** LAMBDA - WAVELENGTH
*****  RAREA - RECEIVING AREA
*****  TAREA - TRANSMITTING AREA
*****TOTAREA - CAPTURE AREA
*****      DZ - FRACTION OF 2*D**2/LAMBDA
*****      ET - EDGE TAPER
*****       K - WAVE NUMBER
*****      NO - FREE SPACE IMPEDANCE
*****NXO NXO - NUMBER OF POINTS IN RECEIVING GRID
*****NXL NXL - NUMBER OF POINTS IN TRANSMITTING GRID
*****       J - SQRT(-1)
*****     KEM - EQUATION CONSTANT
*****      FF - FAR FIELD DISTANCE
*****     EFF - EFFICIENCY
*****    PERC - EFFICIENCY PERCENT
*  DXY & DXY - DELTAS FOR RECEIVING GRID
*  DXY & DXY - DELTAS FOR TRANSMITTING GRID
*****(XO,YO) - POINT IN RECEIVING GRID
*****(XL,YL) - POINT IN TRANSMITTING GRID
*****     PR1 - POWER TRANSMITTED
C****     PR2 - POWER RECEIVED
*****     ETX - TRANSMITTED ELECTRIC FIELD
*   TREX ERX - RECEIVED ELECTRIC FIELD
*   TRMY HRY - RECEIVED MAGNETIC FIELD
******* VARIABLE DECLARATION *****
      INTEGER ANO, BNO, CNL, DNL
      REAL LAMBDA,LF,M,KEM,K,NO
      COMPLEX ERX(300,300),HRY(300,300),ETX(300,300),TREX,TRMY,J
****************************
***** ***** INPUT DATA *****

****************************
           LAMBDA  =   .1
           LF      =   .5
           TAREA   =  0.7854
           RAREA   =  0.7854
           TOTAREA =  1.5708
           DZ      =   .1
           ET      =  0.
C**** *** CONSTANTS ***
           J      = (0.,1.)
           PI     = 4 * ATAN(1.)
           NO     = 120 * PI
           XAL    = SQRT(TAREA) / 2
           XAO    = SQRT(RAREA) / 2
           XAT    = SQRT(TOTAREA) / 2
           K      = 2. * PI / LAMBDA
           KEM    = 1. / (4. * PI)
           DXY    = LAMBDA * LF
           NXL    = 2. * XAL / DXY + 1.1
           NXO    = 2. * XAO / DXY + 1.1
```

```
          NXT     = 2. * XAT / DXY + 1.1
          M       = 10**(-ET / 20)
          FF      = (2. * (2. * XAL)**2. / LAMBDA) * DZ
C****
C****
C****
C**** *** LOOP FOR TRANSMITTING GRIDS ***

      DO 50 CNL = 1, NXL
           XL = -XAL + DXY * (CNL - 1)
           DO 25 DNL = 1, NXL
                YL = -XAL + DXY * (DNL - 1)
C**** *** TRANSMITTED ELECTRIC FIELD ***
                ETX(CNL,DNL) = M + (1. - M) * (1. - (XL**2 + YL**2)/
     1                           (XAL**2))**2
 25        CONTINUE
 50   CONTINUE
C****
C****
C****
C**** *** LOOP FOR RECEIVING GRIDS ***
C****
C****
C****
      DO 200 ANO = 1, NXT
           XO = -XAT + DXY * (ANO - 1)
           DO 100 BNO = 1, NXT
                YO = -XAT + DXY * (BNO - 1)
C****
C****
C**** ***** SUBROUTINE FOR RECEIVED FIELDS CALCULATION *****
C****
C****
C****
                CALL EANDH(M,K,J,NXL,XAL,DXY,FF,NO,KEM,
     1                     TREX,TRMY,ETX,XO,YO)
                ERX(ANO,BNO) = TREX
                HRY(ANO,BNO) = TRMY
                TREX = 0
                TRMY = 0
 100       CONTINUE
 200  CONTINUE

C****
C****
C**** ***** SUBROUTINE FOR POWER AND EFFICIENCY CALCULATION *****
C****
C****
C****
      CALL FINISH(ERX,HRY,M,NXL,NXO,ET,ETX,KEM,LF,RAREA,TAREA,
     1            TOTAREA,XAL,XAO,DXY,DZ,XO,YO,NO,FF,NXT,XAT)
      STOP
      END
C****
C**** *******************************************************************
C****
C**** THIS SUBROUTINE CALCULATES THE XYZ COMPONENTS OF THE RECEIVED
C**** ELECTRIC AND MAGNETIC FIELDS FOR EVERY POINT IN THE RECEIVING
C**** APERTURE BY ADDING THE EFFECT OF EVERY POINT IN THE TRANSMITTING
C**** APERTURE.
```

```
C****
C**** *****************************************************************
      SUBROUTINE EANDH(M,K,J,NXL,XAL,DXY,FF,NO,KEM,TREX,TRMY,ETX,XO,YO)
      REAL M,KEM,K,NO
      INTEGER CNL,DNL
      COMPLEX J,PHASOR,EFOX,ESOX,ETX(300,300),REX,RMY,TREX,TRMY
C**** *** LOOP FOR TRANSMITTING GRIDS ***
      DO 200 CNL = 1, NXL
           XL = -XAL + DXY * (CNL-1)
           DO 100 DNL = 1, NXL
                YL = -XAL + DXY * (DNL - 1)
C****
C****
C**** *** DISTANCE ***
C****
C****

                R = SQRT((XO-XL)**2 + (YO-YL)**2 + FF**2)
C****
C**** *** COS X ***
C****
C****
                COSX = FF / R
C****
C****
C****
C**** *** FIRST ORDER RECEIVED ELECTRIC FIELD ***
C****
C****
C****
                PHASOR = (COS(K*R) - J*SIN(K*R)) / R
                EFOX   = J * K * ETX(CNL,DNL) * COSX * PHASOR
C****
C****
C**** *** SECOND ORDER RECEIVED ELECTRIC FIELD ***
C****
C****
                PHASOR = (COS(K*R) - J*SIN(K*R)) / R**2
                ESOX   = ETX(CNL,DNL) * COSX * PHASOR
C****
C****
C****
C**** *** COMBINED RECEIVED ELECTRIC ELECTRIC FIELDS ***
C****
C****
C****
                REX = KEM * (EFOX + ESOX) * DXY * DXY
C**** *** RECEIVED MAGNETIC FIELD ***
                RMY =  REX / NO
C****
C****
C**** *** TOTAL FIELDS ***
C****
C****
C****
                TREX = TREX + REX
                TRMY = TRMY + RMY
 100       CONTINUE

 200  CONTINUE
```

```
      RETURN
      END
C**** *****************************************************************
C****
C**** THIS SUBROUTINE CALCULATES THE TOTAL POWER RECEIVED, TOTAL
C****
C**** POWER TRANSMITTED AND EFFICIENCY PERCENT.
C****
C**** *****************************************************************
C****
C****
C****
      SUBROUTINE FINISH(ERX,HRY,M,NXL,NXO,ET,ETX,KEM,LF,RAREA,TAREA,
     1                  TOTAREA,XAL,XAO,DXY,DZ,XO,YO,NO,FF,NXT,XAT)
      REAL LF,M,KEM,NO
      INTEGER ANO,BNO
      COMPLEX ERX(300,300),HRY(300,300),ETX(300,300)
C****
C**** ***** TOTAL POWER RADIATED *****
C****
C****
      PR1 = 0
      DSZO = DXY * DXY
      DO 200 ANO = 1, NXT
           DO 100 BNO = 1, NXT
                DSZO = DXY * DXY
                PR1 = .5 * REAL(ERX(ANO,BNO) * CONJG(HRY(ANO,BNO)))
     1                * DSZO + PR1
 100       CONTINUE
 200  CONTINUE
C****
C****
C****
C**** ***** TOTAL POWER RECEIVED *****
C****
      PR2 = 0
      N1 = (-XAO + XAT) / DXY + 1.1
      N2 = (XAO + XAT) / DXY + 1.1
      DO 800 ANO = N1, N2
           DO 900 BNO = N1, N2
                PR2 = .5 * REAL(ERX(ANO,BNO) * CONJG(HRY(ANO,BNO)))
     1                * DSZO + PR2
900        CONTINUE
800   CONTINUE
C**** ***** EFICIENCY PERCENT CALCULATION *****
C****
      EFF = PR2 / PR1
      PERC = EFF * 100
C****
C**** *** OUTPUT DATA ***
      WRITE(9,*) 'FRACTION OF 2*D**2/LAMBDA = ',DZ
      WRITE(9,*) 'DISTANCE BETWEEN APERTURES = ',FF/1000,'KM'
      WRITE(9,*) 'EDGE TAPER = ',ET,'DB'
      WRITE(9,*) 'TRANSMITTING APERTURE AREA = ',TAREA,'SQ. MTS'
      WRITE(9,*) 'RECEIVING APERTURE AREA = ',RAREA,'SQ. MTS'
      WRITE(9,*) 'CAPTURE APERTURE AREA = ',TOTAREA,'SQ. MTS'
      WRITE(9,*) 'TOTAL POWER RADIATED = ',PR1,'W'
      WRITE(9,*) 'TOTAL POWER RECEIVED = ',PR2,'W'
      WRITE(9,*) 'EFFICIENCY PERCENT = ',PERC,'%'
      RETURN
```

```
/*EXEC TO RUN BPE*/
SETUP FTN
"FTNLIB"
"FI 09 DISK CAACOSTA BPE a1"
"LOAD beam (start"
```

REFERENCES

1. George Goubau and Felix Schwering, "Free Space Beam Transmission", Microwave Power Transmission Engineering, E. C. OKress, ed. Academic Press, 1968) Vol 1, pp 241-255.

2. Shinya Takeshita, "Power Transfer Efficiency Between Focused Circular Antennas with Gaussian Illumination in Fresnel Region", IEEE Transactions on Antennas and Propagation, May 1968.

3. J. Durnin, J. J. Miceli,Jr., J. H. Eberly, "Diffraction-Free Beams", Physical Review Letters, Volume 58, No. 15, 13 April 1987.

4. J. W. Sherman, "Properties of Focused Apertures in the Fresnel Region", IRE Transactions on Antennas and Propagation, 1962.

5. B. Houshmand and S. W. Lee, "Study of Beam Waveguide Fed Cassegrian Reflector Antennas", Department of Electrical and Computer Engineering, University of Illinois, 1406 W. Green St., Urbana, Il. Report YRS89-1.

6. M. I. Hoffert, G. Miller, B. Heilweil, W. Ziegler and M. Kadiramangalam, "Earth-To-Satellite Microwave Beams: Innovative Approach to Space Powers", SPIE Vol. 873 Microwave and Particle Beam Sources and Propagation 1988.

7. S. Silver, Microwave Antenna Theory and Design. Boston Technical Publisher, Inc., 1964.

8. Balanis, C. A., Antenna Theory Analysis and Design. Haper & Row, 1982.



Figure 1. Geometry for calculating the transmission efficiency



Figure 2. Parameters description for obtaining beam power efficiency



Figure 3. Efficiency curves resulting from equal areas with two different tapers



Figure 4. Efficiency curves resulting from different transmitting and receiving apertures size .



Figure 5. Efficiency curves resulting from different transmitting and receiving apertures size .



Figure 6. Efficiency curves resulting from different transmitting and receiving apertures size .



Figure 7. Efficiency curves for different receiving aperture size at far-field range



Figure 8. Efficiency curve for different transmitting aperture tapers at near-field range.

REPORT DOCUMENTATION PAGE

Form Approved
OMB No. 0704-0188

1. AGENCY USE ONLY *(Leave blank)*

2. REPORT DATE
February 1992

3. REPORT TYPE AND DATES COVERED
Technical Memorandum

4. TITLE AND SUBTITLE
Microwave Beam Power Transmission
at an Arbitrary Range

5. FUNDING NUMBERS
WU-679-40-00

6. AUTHOR(S)
L.R. Pinero, J.L. Christian, Jr., and R.J. Acosta

7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES)
National Aeronautics and Space Administration
Lewis Research Center
Cleveland, Ohio 44135-3191

8. PERFORMING ORGANIZATION REPORT NUMBER
E-6841

9. SPONSORING/MONITORING AGENCY NAMES(S) AND ADDRESS(ES)
National Aeronautics and Space Administration
Washington, D.C. 20546-0001

10. SPONSORING/MONITORING AGENCY REPORT NUMBER
NASA TM-105424

11. SUPPLEMENTARY NOTES
Responsible person, R.J. Acosta, (216) 433-6640. A copy of the program is available from Computer Software Management and Information Center (COSMIC), 382 East Broad Street, Athens, Georgia 30602, (404) 542-3265.

12a. DISTRIBUTION/AVAILABILITY STATEMENT
Unclassified - Unlimited
Subject Category 32

12b. DISTRIBUTION CODE

13. ABSTRACT ***(Maximum 200 words)***
The power transfer efficiency between two circular apertures at an arbitrary range is obtained numerically. Two apertures can have generally different sizes and arbitrary taper illuminations. The effect of distance and taper illumination on the transmission efficiency are investigated for equal size apertures. The result shows that microwave beam power is more effective at close ranges, namely distances less than $2D^2$/lambda. Also shown was the power transfer efficiency increase with taper illumination for close range distances. A computer program was developed for calculating the power transfer efficiency at an arbitrary range.

14. SUBJECT TERMS
Antenna; Microwave power; Power transfer efficiency

15. NUMBER OF PAGES
18

16. PRICE CODE
A03

17. SECURITY CLASSIFICATION OF REPORT	18. SECURITY CLASSIFICATION OF THIS PAGE	19. SECURITY CLASSIFICATION OF ABSTRACT	20. LIMITATION OF ABSTRACT
Unclassified	Unclassified	Unclassified	

National Aeronautics and
Space Administration

Lewis Research Center
Cleveland, Ohio 44135

NASA